
Abstract12

It is nowadays customary to organize libraries of machine checked proofs around hierarchies of13

algebraic structures [2, 6, 8, 16, 18, 23, 27]. One influential example is the Mathematical Components14

library on top of which the long and intricate proof of the Odd Order Theorem could be fully15

formalized [14].16

Still, building algebraic hierarchies in a proof assistant such as Coq [9] requires a lot of manual17

labor and often a deep expertise in the internals of the prover [13, 17]. Moreover, according to our18

experience [26], making a hierarchy evolve without causing breakage in client code is equally tricky:19

even a simple refactoring such as splitting a structure into two simpler ones is hard to get right.20

In this paper we describe HB, a high level language to build hierarchies of algebraic structures21

and to make these hierarchies evolve without breaking user code. The key concepts are the ones of22

factory, builder and abbreviation that let the hierarchy developer describe an actual interface for23

their library. Behind that interface the developer can provide appropriate code to ensure backward24

compatibility. We implement the HB language in the hierarchy-builder addon for the Coq system25

using the Elpi [11, 28] extension language.26

1 Introduction34

Modern libraries of machine checked proofs are organized around hierarchies of algebraic35

structures [2, 6, 8, 16, 18, 23, 27]. For example the Mathematical Components library for36

the Coq system [9] provides a very rich, ever growing, hierarchy of structures such as group,37

ring, module, algebra, field, partial order, order, lattice. . . The hierarchy does not only serve38

the purpose of organizing knowledge, but also to make it easy to exploit it. Indeed the39

interactive prover can take advantage of the structure of the library and the relation between40

its concepts to infer part of information usually left implicit by the user, a capability that41

turned out to be key to tame the complexity and size of the formal proof of the Odd Order42

Theorem [14].43

8:2 Hierarchy Builder

The hierarchy of the Mathematical Components library is implemented following the dis-44

cipline of Packed Classes initially introduced by Garillot, Gonthier, Mahboubi and Rideau [13]45

and later also adopted by Affeldt, Nowak, and Saikawain to describe a hierarchy of monadic46

effects [2] and by Boldo, Lelay, and Melquiond in the Coquelicot library of real analysis [6].47

We call Packed Classes a discipline, and not a language, because, in spite of its many virtues,48

it is unwieldy to use. In particular it leaks to the user many of the technical details of49

the Coq system. As a result, one needs to be a Coq expert in order to build or modify a50

hierarchy, and even experts make mistakes as shown in [26]. Another inconvenience of the51

Packed Classes discipline is that even simple changes to the hierarchy, such as splitting a52

structure into two simpler ones, break user code.53

In this paper we describe HB, a high level language to build hierarchies of algebraic54

structures and to make these hierarchies evolve without breaking user code. The key concepts55

are the ones of factory, builder and abbreviation that let the hierarchy developer describe an56

actual interface for their library. Behind that interface the developer can provide appropriate57

code to ensure backward compatibility. We implement the HB language by compiling it to a58

variant of the Packed Classes discipline, that we call flat, in the hierarchy-builder addon59

for Coq. We write this addon using the Elpi [11, 28] extension language.60

61

To sum up, the main contributions of the paper are:62

the design of the HB language,63

the compilation of HB to the (flat) Packed Classes discipline, and64

the implementation of HB in the hierarchy-builder addon for Coq.65

The paper is organized as follows. Via an example we introduce HB and its key ideas. We66

then describe the discipline of Packed Classes and we show how HB can be compiled to it.67

We then discuss the implementation of the Coq addon via the Elpi extension language and68

we position HB in the literature.69

2 HB by examples: building and evolving a hierarchy70

The first version of our hierarchy (that we name V1) features only two structures: Monoid71

and Ring. We use notations from Appendix A to define them.72

1 HB.mixin Record Monoid_of_Type M := {
2 zero : M;
3 add : M -> M -> M;
4 addrA : associative add; (* `add` is associative. *)
5 add0r : left_id zero add; (* `zero` is the left and right neutral *)
6 addr0 : right_id zero add; (* element with respect to `add`. *)
7 }.
8 HB.structure Definition Monoid := { M of Monoid_of_Type M }.
9

10 HB.mixin Record Ring_of_Monoid R of Monoid R := {
11 one : R;
12 opp : R -> R;
13 mul : R -> R -> R;
14 addNr : left_inverse zero opp add; (* `opp x` is the left and right additive *)
15 addrN : right_inverse zero opp add; (* inverse of `x`. *)
16 mulrA : associative mul; (* `mul` is associative. *)
17 mul1r : left_id one mul; (* `one` is the left and right neutral *)
18 mulr1 : right_id one mul; (* element with respect to `mul`. *)
19 mulrDl : left_distributive mul add; (* `mul` is left and right distributive *)
20 mulrDr : right_distributive mul add; (* over `add`. *)
21 }.
22 HB.structure Definition Ring := { R of Monoid R & Ring_of_Monoid R }.

In order to build a structure we need to declare some factories and later assemble them.73

One kind of factory supported by HB, the simplest one, is called mixin and is embodied by74

a record that gathers operations and properties.75

Mixins are declared via the HB.mixin command that takes a record declaration with a76

type parameter and a possibly empty list of factories for that type. The code between lines 177

and 7 declares a mixin that can turn a naked type M into a monoid, hence we chose the name78

to be Monoid_of_Type.79

The HB.structure command takes in input a definition for a sigma type S that equips a80

type with a list of factories. It registers a structure S in the hierarchy placing any definition81

specific to that structure inside a Coq module named S. Line 8 hence forges the Monoid82

structure.83

Line 10 declares a second mixin collecting the operations and properties that are needed84

in order to enrich a monoid to a ring, hence the name Ring_of_Monoid. Indeed this time the85

type variable R is followed by Monoid that enriches R with the operations and properties of86

monoids. As a consequence add and zero can be used to express the new properties.87

The last line declares the Ring structure to hold all the axioms declared so far. We can88

now inspect the contents of the hierarchy and then proceed to build a theory about abstract89

rings, register examples (instances) of ring structures and finally use the abstract theory on90

these examples.91

1 Print Monoid.type. (* Monoid.type := { sort : Type; ... } *)
2 Check @add. (* add : forall M : Monoid.type, M -> M -> M *)
3 Check @addNr. (* addNr : forall R : Ring.type, left_inverse zero opp add *)
4

5 Lemma addrC {R : Ring.type} : commutative (@add R).
6 Proof. (* Proof by Hankel 1867, in Appendix B *) Qed.
7

8 Definition Z_Monoid_axioms : Monoid_of_Type Z :=
9 Monoid_of_Type.Build Z 0%Z Z.add Z.add_assoc Z.add_0_l Z.add_0_r.

10

11 HB.instance Z Z_Monoid_axioms.
12

13 Definition Z_Ring_axioms : Ring_of_Monoid Z :=
14 Ring_of_Monoid.Build Z 1%Z Z.opp Z.mul
15 Z.add_opp_diag_l Z.add_opp_diag_r Z.mul_assoc Z.mul_1_l Z.mul_1_r
16 Z.mul_add_distr_r Z.mul_add_distr_l.
17

18 HB.instance Z Z_Ring_axioms.
19

20 Lemma exercise (m n : Z) : (n + m) - n * 1 = m.
21 Proof. by rewrite mulr1 (addrC n) -(addrA m) addrN addr0. Qed.

We can print the type for monoids as forged by HB (line 1). It packs a carrier, called sort,92

and the collection of operation and properties that we omit for brevity. We can also look93

at the type of two constants synthesized by HB out of the hierarchy declaration. Remark94

that while the names of the constants come from the names of mixin fields, their types differ.95

In particular they are quantified over a Monoid.type or Ring.type, and not a simple type96

as in the mixins. Moreover we evince that the sort projection is declared as an implicit97

coercion [24] and is automatically inserted in order to make M -> M -> M a meaningful type98

for binary operations on the carrier of M. Last, we see that properties are quantified on (hence99

apply to) the structure they belong to but use, in their statements, operations belonging to100

simpler structures. For example addNr is a property of a ring but its statement mentions add,101

the operation of the underlying monoid.102

We then follow the proof of Hankel [5] to show that ring axioms imply the commutativity103

of the underlying monoid (line 5). This simple example shows we can populate the theory of104

abstract rings with new results.105

We use the Monoid_of_Type.Build abbreviation (line 8) in order to build an instance of106

the Monoid structure for binary integers Z. We then register that monoid instance as the107

canonical one on Z (line 11) via the command HB.instance. We can similarly declare that Z108

forms a ring by using the Ring_of_Monoid.Build abbreviation (lines 13 and 18). Note that109

the Ring_of_Monoid.Build abbreviation is not a plain record constructor for Ring_of_Monoid,110

since that would require more arguments, namely the monoid ones (see the _ at line 13).111

The abbreviation synthesized by HB infers them automatically (as in [17, Section 7]) thanks112

to the HB.instance declaration given just above.113

From now on the axioms as well as the abstract theory of rings apply to integers, as shown114

in lemma exercise. The details of the proof do not matter here, what is worth pointing115

out is that in a single statement we mix monoid (e.g. +) and ring (e.g. -) operations and116

in the proof we use monoid axioms (e.g. addrA), ring axioms (e.g. addrN) and ring lemmas117

(e.g. addrC), all seamlessly.118

2.1 Evolution of the hierarchy119

We proceed by accommodating the intermediate structure of Abelian groups.120

1 HB.mixin Record Monoid_of_Type M := { ... (* unchanged *) ... }.
2 HB.structure Definition Monoid := { M of Monoid_of_Type M }.
3

4 HB.mixin Record AbelianGroup_of_Monoid A of Monoid A := {
5 opp : A -> A;
6 addrC : commutative (add : A -> A -> A);
7 addNr : left_inverse zero opp add;
8 }.
9 HB.structure Definition AbelianGroup := { A of Monoid A & AbelianGroup_of_Monoid A }.

10

11 HB.mixin Record Ring_of_AbelianGroup R of AbelianGroup R := {
12 one : R;
13 mul : R -> R -> R;
14 mulrA : associative mul;
15 mul1r : left_id one mul; mulr1 : right_id one mul;
16 mulrDl : left_distributive mul add; mulrDr : right_distributive mul add;
17 }.
18 HB.structure Definition Ring := { R of AbelianGroup R & Ring_of_AbelianGroup R }.
19

20 Lemma addrN {A : AbelianGroup.type} : right_inverse zero opp add.
21 Proof. by move=> x; rewrite addrC addNr. Qed.

Some operations and properties were moved from the old mixin for rings into a newborn121

mixin AbelianGroup_of_Monoid that gathers the axioms needed to turn a monoid into an122

Abelian group. Consequently the mixin for rings is now called Ring_of_AbelianGroup (instead123

of Ring_of_Monoid) since it expects the type R to be already an Abelian group and hence124

gathers fewer axioms. While operations moved from one structure to another, some properties125

undergo a deep change in their status. The lemma addrC part of the abstract theory of rings126

is now an axiom of Abelian groups, while addrN is no more an axiom of rings, but rather a127

theorem of the abstract theory of Abelian groups proved at line 20.128

It is worth clarifying here what axioms and lemmas are by looking at the two distinct129

use cases for a hierarchy: 1) develop the abstract theory of some structure; 2) apply the130

abstract theory on some concrete example. In the first case all the axioms of the structure131

are (assumed to be) theorems so the distinction between axiom and theorem does not matter.132

This is what happens in the proof of addNr by taking R to be of type AbelianGroup.type as an133

assumption. In the second case, in order to show Coq that a data type and some operations134

forms a structure, one must pick the axioms of the structure an prove them for that specific135

type and operations. For example the Ring_of_Monoid.Build abbreviation is used at page 3136

to package the set of proofs that make Z a ring (proviso Z is already a monoid).137

With this new version of the hierarchy, that we name V2, the axiomatic that was previously138

exposed to user changed, and indeed code written for version V1 breaks. For example the139

declaration of the canonical ring over the integers fails, if only because we do not have a140

Ring_of_Monoid.Build abbreviation anymore.141

Our objective is to obtain a version of the hierarchy, that we name V3, that does not142

only feature Abelian groups but that is also backward compatible with V1.143

2.2 The missing puzzle piece144

Figure 1 The evolution of the hierarchy. V3 is backward compatible with V1, while V2 is not.

The key to make a hierarchy evolve without breaking user code is the full fledged notion145

of factory (the mixins seen so far are degenerate, trivial, factories). Factories, like mixins,146

are packages for operations and properties but are not directly used in the definition of147

structures. Instead a factory is equipped with builders: user provided pieces of code that148

extract from the factory the contents of mixins, so that existing abbreviations can be used.149

As depicted in Figure 1 we change again the hierarchy by declaring a Ring_of_Monoid fact-150

ory, that, from the user point of view, will look indistinguishable from the old Ring_of_Monoid151

mixin and hence grant backward compatibility between version V3 and version V1.152

1 HB.factory Record Ring_of_Monoid R of Monoid R := { ... (* unchanged *) ...}.
2

3 HB.builders Context R (f : Ring_of_Monoid R).
4

5 Lemma addrC : commutative add. Proof. (* The same proof as before *) Qed.
6

7 Definition to_AbelianGroup_of_Monoid :=
8 AbelianGroup_of_Monoid.Build R opp addrC addNr. (* addrN unused *)
9 HB.instance R to_AbelianGroup_of_Monoid.

10

11 Definition to_Ring_of_AbelianGroup := Ring_of_AbelianGroup.Build R one mul
12 mulrA mul1r mulr1 mulrDl mulrDr.
13 HB.instance R to_Ring_of_AbelianGroup.
14

15 HB.end.

FSCD 2020



8:6 Hierarchy Builder

The record Ring_of_Monoid is the same we declared as a mixin in version V1. In order to153

make a factory out of it we equip it with two definitions that embody the builders. The first154

is to_AbelianGroup_of_Monoid and explains how to build an AbelianGroup structure out of the155

factory axioms (named f, line 3). This construction is also registered as canonical for R, so156

that the next construction to_Ring_of_AbelianGroup can call the Ring_of_AbelianGroup.Build157

abbreviation that requires R to be an AbelianGroup. It is worth pointing out that the proof158

of addrC we had in V1 is now required in order to write the builder for Abelian groups, while159

the addrN field is not used (the same statement is already part of the theory of Abelian160

groups, see line 20 of the previous code snippet).161

Thanks to this factory we can now declare Z to be an instance of a ring using the162

Ring_of_Monoid.Build abbreviation. The associated builders generate, behind the scenes,163

instances of the Ring_of_AbelianGroup and AbelianGroup_of_Monoid mixins that in turn are164

used to build instances of the AbelianGroup and Ring structures. Indeed, when used in the165

context of the hierarchy version V3, the command HB.instance Z Z_Ring_axioms makes Z an166

instance of both structures, and not just the Ring one as in version V1. Thanks to that the167

proof of example can use the theory of both structures, for example addrC holds on Abelian168

groups, addrA holds on monoids, while addr1 holds on rings. As a result the very same proof169

works on both version V1 and V3.170

Last, it is worth pointing out that the new factory makes the following two lines equivalent171

(the former declares rings in V1) since they both describe the same set of mixins.172

1 HB.structure Definition Ring := { R of Monoid R & Ring_of_Monoid R }.
2 HB.structure Definition Ring := { R of AbelianGroup R & Ring_of_AbelianGroup R }.

This is another example of client code that would not break: the client of the hierarchy173

is allowed to declare new structures on top of existing ones.174

2.3 HB in a nutshell175

By using HB the hierarchy designer has the following freedoms and advantages.176

Operations and properties (axioms) are made available to the user as soon as they are177

used in a structure. The hierarchy designer is free to move them from one to another and178

replace an axiom by a lemma and vice versa.179

Structures cannot disappear but the way they are built may change. The hierarchy180

designer is free to split structures into smaller ones in order to better factor and reuse181

parts of the hierarchy and the library that follows it.182

Mixins cannot disappear but can change considerably in their implementation. A mixin183

can become a full fledged factory equipped with builders to ensure backward compatibility.184

HB high level commands compile to the discipline of Packed Classes (Coq modules, records,185

coercions, implicit arguments, canonical structure instances, notations, abbreviations).186

This process lifts a considerable burden from the shoulders of the hierarchy designer and187

the final user who are no longer required to master the details of Packed Classes.188

3 The HB language189

The Coq terms handled byHB are subdivided in five categories, the mixinsM, the factories F ,190

builders B, the classes C and the structures S. Mixins, factories and instances are tagged191

by the user, through the commands HB.mixin, HB.factory and HB.instance, and the user192

may rely on their implementation. However structures and classes are generated with the193

command HB.structure and builders are generated when using HB.instance while declaring194

a factory, and the user may only refer to structure types, but should never rely on their195

implementation, neither should they rely on explicit builders.196

3.1 Mixins, factories and instances197

In this section, we call “distinguished” a Coq definition or record that the developer of a198

library has labeled “mixin”, “factory”, “builder”, “class” or “structure”.199

I Definition 1 (M, mixins). A mixin m ∈M is a distinguished Coq record with one or more200

parameters. The first parameter must be a (T : Type), while the other parameters are mixins201

(mi)i∈{1,...,n}, each of which is applied to T and possibly previous mixin variables. I.e.202

1 Record m (T : Type)
−−−−−−−−−→
(p : m T pσ)n : Type := { .. }.

where
−−−−−−−−−→
(p : m T pσ)n = (p1 : m1 T ) . . .

(
pn : mn T pσ(n,1) . . . pσ(n,qn)

)
and where for all203

i ∈ {1, . . . , n} we have qi ∈ N and the arguments of mi ∈M consist in qi of the previously204

quantified mixin parameters pk, i.e. for all k ∈ {1, . . . , qi}, we have σ(i, k) ∈ {1, . . . , i− 1}.205

I Definition 2 (dep, mixin dependencies). Given m ∈M, we define dep(m) ∈ ℘(M) as the206

set of all mixins that occur as parameters of m, i.e. dep(m) = {m1, . . . ,mn}.207

I Remark 3. For all i ∈ {1, . . . , n}, we have dep(mi) =
{
mσ(i,1), . . . ,mσ(i,qi)

}
.208

I Definition 4. If f : X → ℘(Y), then f? : ℘(X )→ ℘(Y) is defined as f?(X) =
⋃
x∈X f(x).209

I Proposition 5 (dep is transitively closed and describes a DAG).

∀m ∈M, dep?(dep?(m)) ⊆ dep?(m) and m /∈ dep(m).210

Proof. Indeed dep is transitively closed because records are well typed in the empty context211

and describes a DAG since Coq does not admit circular definitions. J212

I Definition 6 (factories F). A factory f ∈ F is a distinguished Coq record or definition213

with one or more parameters. The first parameter must be a (T : Type), while the other214

parameters are n mixins, applied to T and previous mixin variables. I.e.215

1 Record (* or Definition *) f (T : Type)
−−−−−−−−−→
(p : m T pσ)n : Type := ...

I Definition 7 (requires, factory requirements.). Given f ∈ F , we define requires(f) ∈ ℘(M)216

as the set of all mixins that occur as parameters of f , i.e. dep(f) = {m1, . . . ,mn}.217

The following property holds,218

I Proposition 8 (requires is closed under dep). ∀f ∈ F , dep?(requires(f)) ⊆ requires(f).219

Proof. Because Coq records and global definitions are well typed in the empty context. J220

I Definition 9 (builders B). A builder µ ∈ B is a distinguished function whose return221

type is a mixin mn+1 ∈ M and whose parameters are the carrier type (T : Type), mixins222

{m1, . . . ,mn} ∈ ℘(M) and a factory f ∈ F , such that requires(f) = {m1, . . . ,mn}. I.e.223

1 Definition µ (T : Type)
−−−−−−−−−→
(p : m T pσ)n: f T p1 . . . pn -> mn+1 T pσ(n+1,1) . . . pσ(n+1,qn+1) := ...

Note that the builders of a given factory have the same set of dependencies.224

8:8 Hierarchy Builder

I Definition 10 (from). We define from(f,mn+1) = µ to be the (unique) builder for mn+1225

from the factory f , when it exists.226

Note that from is not a total function, and that from(f,m) is defined if and only if there227

is a declared builder that shows how to build m from f . In this case, we say f provides m.228

I Definition 11 (provides). provides(f) = {m | from(f,m) is defined} ∈ ℘(M) the set of229

mixins that a factory f ∈ F provides, through its builders.230

Mixins are declared by the user as the fundamental building blocks of a hierarchy. As the231

next proposition shows they shall not be regarded as different from regular factories, since232

they are a special case.233

I Proposition 12 (M⊆ F). There is a way to see mixins as factories.234

Proof. For allm ∈M we have requires(m) = dep(m), provides(m) = {m} and from(m,m) =235 (
fun T

−−−−−−−−→
(p : m T pσ)n (x : m T p1 . . . pn) => x

)
∈ B. J236

I Coq command to declare a new mixin:237

1 HB.mixin Record M T of f1 . . . fn := { .. }.

Declares the record axioms inside a module M. This record M.axioms depends on the mixins238

requires(M.axioms) = dep(M.axioms) = provides? ({f1, . . . , fn}) and is registered both as a239

mixin and a factory. Finally It exports an abbreviation M.Build and a notation M standing240

for M.axioms, so that the module name can be used to denote the axioms record it contains.241

I Coq command to declare an instance: HB.instance X b1 . . . bk, synthesizes terms242

corresponding to all the mixins that can be built from the bi. Indeed if bi : fi T . . ., then this243

command creates elements of types provides? ({f1, . . . , fk}) . This command also generates244

unification hints as described in Section 4.245

I Coq commands to declare a new factory and generate new builders:246

1 HB.factory Record F T of f1 . . . fn := { .. }.
2 HB.builders Context T (a : F T).
3

4 Definition bn+1 : fn+1 T .. := ...
5 HB.instance T bn+1.
6 ..
7 Definition bn+k : fn+k T .. := ...
8 HB.instance T bn+k.
9 HB.end.

Declares the record axioms inside a module F. This record F.axioms depends on the mixins247

requires(F.axioms) = provides? ({f1, . . . , fn}) and is registered as a factory. It exports an248

abbreviation F.Build and a notation F standing for F.axioms, so that the module name249

can be used to denote the axioms record it contains. Finally it uses the factory instances250

bn+1, . . . , bn+k provided by the user in order to derive builders, so that provides(F) =251

provides? ({fn+1, . . . , fn+k}) .252

It is thus necessary that requires? ({fn+1, . . . , fn+k}) ⊆ provides? ({f1, . . . , fn}) .253

Note that the bi are not builders since their return types are not necessarily mixins, but254

could be factories. However, since all factories provide mixins through builders, we obtain255

builders out of each bi by function composition.256

3.2 Classes and structures257

I Definition 13 (C, class). A class c ∈ C is a distinguished Coq record with one parameter258

(T : Type). The type of each field is a mixin inM applied to T and, if needed, any number259

of other fields:260

1 Record c (T : Type) := { p1 : m1 T; .. ; pi : mi T pσ(i,1) . . . pσ(i,qi); ..}.

I Definition 14 (def , class definition). We call def (c) ∈ ℘(M) the set of mixins mentioned261

in the fields of the class, i.e. {m1, . . . ,mn}. Given that class records are well typed in the262

empty context the set of mixin records is closed transitively. The implementation enforces263

that no two class records contains the same set of mixins (disregarding the order of the fields),264

i.e. def is injective.265

I Programming invariant for def For all f ∈ F :266

1. ∃c ∈ C, def (c) = requires?(f) ∪ provides?(f),267

2. ∃C ⊆ C, def ?(C) = requires?(f).268

Classes could also be seen as factories.269

I Proposition 15 (C ⊆ F). For all m ∈ def (c) we have from(c,mi) = pi, and it follows that270

requires(c) = ∅, provides(c) = def (c).271

However since classes are generated, their implementation may change, thus users should272

not rely on constructors of classes. Hence the only way users may refer to a class is as an273

argument of HB.mixin, HB.factory or HB.structure.274

I Definition 16 (S Structure). A structure s ∈ S is a distinguished dependent pair: a Coq275

record where the value of the first field occurs in the type of the second. The first field is276

(sort : Type) and the second field is (class : c sort) for some c ∈ C. As a consequence277

structures are in bijection with classes.278

I Coq command to declare a class and structure: HB.structure Definition M := { A of279

f1 . . . fn } crafts a class M.class_of ∈ C where def (c) = provides? ({f1, . . . , fn}) and the280

corresponding structure M.type ∈ S, together with unification hints as described in Section 4.281

I Definition 17 (≤ ∈ C × C, subclass). c1 ≤ c2 iff def (c2) ⊆ def (c1)282

3.3 Automatic inference of mixins283

Since mixins may change but factories stay the same, HB commands must never rely on a284

particular set of mixins as arguments, and factory arguments must never be given explicitly285

by the user. As described in Sections 3.1 and 3.2, HB commands take a list of factories as286

arguments, which they expand into lists of mixins behind the scene. However factory types287

and constructors have mixin arguments that must be inferred automatically when used. To288

this end, the commands HB.mixin and HB.factory generate abbreviations for the user to289

replace uses of constructors of factories. These commands first create a record faux with a290

constructor Faux and then create abbreviations f and F that automatically fill the mixin291

arguments of faux and Faux respectively. See [17, Section 7] for a detailed description of how292

to implement these abbreviations in Coq.293

1 Record faux T
−−−−−−−−−→
(p : m T pσ)n := Faux { ... }.

2 Notation f T := (faux T ...(* p1 . . . pn inferred when T is known *) ).
3 Notation F T x1 . . . xk := (Faux T ...(* p1 . . . pn inferred when T is known *) x1 . . . xk).
4 Definition b : f T := F T x1 . . . xn.

4 The target language: Coq with Packed, flat, Classes294

The language of Packed Classes [13] is used directly to describe the algebraic hierarchy of the295

Mathematical Components library. It is based on a disciplined use of records and projections296

and on the possibility of extending the elaborator of Coq via the declaration of Canonical297

Structures [17] instances. In this section, we describe the flat variant of Packed Classes, the298

target language of hierarchy-builder, through the hierarchy V3 extended with semirings.299

The hierarchy-builder addon can generate all the Coq declarations in this section300

automatically, but some details are omitted for brevity in this section.301

4.1 Describing structures with records and projections302

We describe mathematical structures with three kinds of dependent records: mixins, classes,303

and structures. As shown in Section 2, a mixin gathers operators and axioms newly introduced304

by a structure. As in [13, Section 2.4][26, Section 2], a class record is parametrized by the305

carrier type (T : Type) and gathers all the operators and axioms of a structure by assembling306

mixins, and a Structure type (a record) bundles a carrier and its class instance, as follows.307

1 Module Monoid.
2 Record axioms (M : Type) : Type :=
3 Class { Monoid_of_Type_mixin : Monoid_of_Type M; }.
4 Structure type : Type := Pack { sort : Type; class : axioms sort; }.
5 End Monoid.

The Monoid module plays the role of a name space and forces us to write qualified names such308

as Monoid.type; as a consequence we can reuse the same unqualified names for other structures,309

i.e., class and structure record can always be named as axioms and type respectively.310

Mixins and classes are internal definitions to structures; in contrast, Monoid.type is part311

of the interface of the monoid structure. As seen in section 2, we declare Monoid.sort as312

an implicit coercion and lift monoid operators and axioms from projections for the monoid313

mixin to definitions and lemmas for Monoid.type as follows.314

1 Coercion Monoid.sort : Monoid.type >-> Sortclass.
2

3 Definition zero {M : Monoid.type} : M :=
4 Monoid_of_Type.zero M (Monoid.Monoid_of_Type_mixin M (Monoid.class M)).
5 Definition add {M : Monoid.type} : M -> M -> M :=
6 Monoid_of_Type.add M (Monoid.Monoid_of_Type_mixin M (Monoid.class M)).
7 Lemma addrA {M : Monoid.type} : associative (@add M).
8 (* Two monoid axioms `add0r` and `addr0` are omitted. *)

Next we define Abelian groups. Since the monoid structure is the bottom of this hierarchy315

its class record Monoid consists of just one mixin. In contrast the class record of Abelian316

groups consists of two mixins where the second one depends on the former (since Abelian317

groups inherit from monoids).318

1 Module AbelianGroup.
2 Record axioms (A : Type) : Type := Class {
3 Monoid_of_Type_mixin : Monoid_of_Type A;
4 AbelianGroup_of_Monoid_mixin : AbelianGroup_of_Monoid A Monoid_of_Type_mixin; }.
5 Structure type : Type := Pack { sort : Type; class : axioms sort }.
6 End AbelianGroup.

In the flat variant of Packed Classes, a class record gathers mixins as its fields directly as319

in above. In contrast, a class record of the non-flat variant packs a class instance of one of320

the superclasses with remaining mixins, which reduces the amount of code to implement321



C. Cohen and K. Sakaguchi and E. Tassi 8:11

inheritance significantly. Since we do not need to care about the amount of code in automated322

generation, hierarchy-builder uses the flat variant of Packed Classes as its target language.323

As in the monoid structure, we declare AbelianGroup.sort as an implicit coercion and lift324

additive inverse opp and Abelian group axioms. Since Abelian groups inherit from monoids,325

we also declare an implicit coercion from Abelian groups to monoids. A coercion between326

structures can be defined in two steps: first a coercion between the class record of the327

superclass to the class record of the subclass (line 2); and a coercion between structure328

records (line 4) relying on the first one.329

1 Coercion AbelianGroup.sort : AbelianGroup.type >-> Sortclass.
2 Coercion AbelianGroup_class_to_Monoid_class (A : Type) (c : AbelianGroup A) :
3 Monoid A := Monoid.Class A (AbelianGroup.Monoid_of_Type_mixin A c).
4 Coercion AbelianGroup_to_Monoid (A : AbelianGroup.type) : Monoid.type :=
5 Monoid.Pack (AbelianGroup.sort A) (AbelianGroup.class A).
6

7 Definition opp {A : AbelianGroup.type} : A -> A := ... .
8 (* Two Abelian group axioms `addrC` and `addNr` are omitted. *)

Generally, a coercion from a structure X to another structure Y can (and should) have the330

following form, thanks to the corresponding coercion between classes X_class_to_Y_class.331

1 Coercion X_to_Y (T : X.type) : Y.type :=
2 Y.Pack (X.sort T) ((* X_class_to_Y_class _ *) (X.class T)).

4.2 Multiple inheritance332

Figure 2 V4 introduces multiple in-
heritance. For brevity we omit the
factories/builders for the upper arrows.

In order to introduce multiple inheritance, we extend333

the hierarchy described in Section 2.2 with the struc-334

ture of semirings as depicted in Figure 2, and name335

it V4. Semirings introduce the binary multiplication336

operator mul and its identity element one. For the337

sake of completeness, the full code of V4 is available338

in Appendix C.339

1 HB.mixin Record SemiRing_of_Monoid S
2 of Monoid S := {
3 one : S;
4 mul : S -> S -> S;
5 (* 7 axioms are omitted. *)
6 }.
7

8 HB.factory Record Ring_of_AbelianGroup R
9 of AbelianGroup R := {

10 (* 2 operators and 5 axioms are omitted. *)
11 }.
12 HB.builders Context R (f : Ring_of_AbelianGroup R).
13 ..
14 HB.end.

Since semirings and Abelian groups do not inherit from each other, the definition of340

semirings and Abelian groups are quite similar:341

1 Module SemiRing.
2 Record axioms (S : Type) : Type := Class {
3 Monoid_of_Type_mixin : Monoid_of_Type S;
4 SemiRing_of_Monoid_mixin : SemiRing_of_Monoid S Monoid_of_Type_mixin;
5 }.
6 Structure type : Type := Pack { sort : Type; class : axioms sort; }.
7 End SemiRing.

We define implicit coercions from the semiring structure to the carrier and the monoid342

structure, and we lift semiring operators and axioms as follows:343

1 Coercion SemiRing.sort : SemiRing.type >-> Sortclass.
2 Coercion SemiRing_to_Monoid : SemiRing.type >-> Monoid.type.
3

4 Definition one {S : SemiRing.type} : S := ... .
5 Definition mul {S : SemiRing.type} : S -> S -> S := ... .
6 (* 7 semiring axioms are omitted. *)

The class record is defined by gathering the monoid, Abelian group, and semiring mixins.344

Since the rings inherit from monoids, semirings, and Abelian groups, we define implicit345

coercions from the ring structure to those three structures.346

1 Module Ring.
2 Record axioms (R : Type) : Type := Class {
3 Monoid_of_Type_mixin : Monoid_of_Type R;
4 SemiRing_of_Monoid_mixin : SemiRing_of_Monoid R Monoid_of_Type_mixin;
5 AbelianGroup_of_Monoid_mixin : AbelianGroup_of_Monoid R Monoid_of_Type_mixin; }.
6 Structure type : Type := Pack { sort : Type; class : axioms sort; }.
7 End Ring.
8

9 Coercion Ring.sort : Ring.type >-> Sortclass.
10 Coercion Ring_to_Monoid : Ring.type >-> Monoid.type.
11 Coercion Ring_to_SemiRing : Ring.type >-> SemiRing.type.
12 Coercion Ring_to_AbelianGroup : Ring.type >-> AbelianGroup.type.

4.3 Linking structures and instances via Canonical Structures347

We recall here how to use the Canonical Structures [17, 25] mechanism, which lets the user348

extend the elaborator of Coq, in order to handle inheritance and inference of structure [1,349

26, 13]. This software component takes as input a term as written by the user and has to350

infer all the missing information that is necessary in order to make the term well typed.351

A first example of elaboration that requires canonical structures is 0 + 1. After removing352

all syntactic facilities, the underlying Coq term is (add _ (zero _)) (one _) where _ stands353

for an implicit piece of information to be inferred and the constants add and zero come from354

the monoid structure while one comes from semirings. When the term is type checked each _355

is replaced by a unification variable written ?v. Respectively, the head and the argument of356

the top application can be typed as follows:357

1 add ?M (zero ?M) : Monoid.sort ?M -> Monoid.sort ?M
2 one ?SR : SemiRing.sort ?SR

where ?M : Monoid.type and ?SR : SemiRing.type. In order to type check the applica-358

tion Coq has to unify Monoid.sort ?M with SemiRing.sort ?SR, which is not trivial: it359

amounts at finding a structure that is both a monoid and a semiring, possibly the smal-360

lest one [26, Sect. 4]. This piece of information can be inferred from the hierarchy and361

its inheritance relation (Definition 17) and we can tell Coq to exploit it by declaring362

SemiRing_to_Monoid : SemiRing.type -> Monoid.type as canonical. With that hint Coq will363

pick ?M to be SemiRing_to_Monoid ?SR as the canonical solution for this unification problem.364

In general, all the coercions between structures must be declared as canonical.365

Another example of elaboration problem is −1, which hides the term opp _ (one _). Here366

opp and one are respectively from Abelian groups and semirings, which do not inherit each367

other but whose smallest common substructure is rings; thus we have to extend the unifier368

to solve a unification problem AbelianGroup.sort ?AbG = SemiRing.sort ?SR by instantiating369

?AbG and ?SR with Ring_to_AbelianGroup ?R and Ring_to_SemiRing ?R respectively where ?R is370

a fresh unification variable of type Ring.type. This hint can be given as follows:371

1 Canonical AbelianGroup_to_SemiRing (S : Ring.type) :=
2 SemiRing.Pack (AbelianGroup.sort (Ring_to_AbelianGroup S)) (Ring.class S)).

Similarly, one can apply an algebraic theory to an instance (an example) of that structure,372

e.g., 2× 3 where 2 and 3 have type Z. The same mechanism of canonical structures let us373

extend the unifier to solve SemiRing.type ? = Z.374

5 The implementation of HB in Coq-Elpi375

The implementation is based on the Elpi extension language for Coq. In this section we376

introduce the features of the programming language that came in handy in the development377

of HB and comment a few code snippets.378

Coq-Elpi [28] is a Coq plugin embedding Elpi and providing an extensive, high level,379

API to access and script the Coq system at the vernacular level. This API lives in the380

coq namespace and lets one easily declare records, coercions, canonical structures, modules,381

implicit arguments, etc. The most basic Coq data type exposed to Elpi is the one of references382

to global declarations:383

1 kind gref type. % The data type of references to global terms
2 type indt inductive -> gref. % eg: Coq.Init.Datatypes.nat
3 type indc constructor -> gref. % eg: Coq.Init.Datatypes.O
4 type const constant -> gref. % eg: Coq.Init.Peano.plus

The arguments of the three constructors are opaque to Elpi, that can only use values of384

these types via dedicated APIs. For example the API for declaring an inductive type will385

generate a value of type inductive that is printed as, for example, «nat».386

Elpi [11] is a dialect of λProlog [19], an higher order logic programming language that387

makes it easy to manipulate abstract syntax tree with binders. Coq-Elpi takes full advantage388

of this capability by representing Coq terms in Higher Order Abstract Syntax [20] style,389

reusing the binder of the programming language in order to represent Coq’s ones. Here is an390

excerpt of the data type of Coq terms:391

1 kind term type. % The data type of Coq terms
2 type global gref -> term. % eg: nat, O, S, plus, ...
3 type fun term -> (term -> term) -> term. % eg: fun x : t => b(x)
4 type app list term -> term. % eg: app [hd|args]
5 ... % all other term constructors are omitted for brevity

Note that the fun constructor holds a λProlog function. In this syntax the Coq term392

(fun x : nat => x x) becomes (fun (global (indt «nat»)) x\ app[x, x]) where x\ binds393

x in the body of the function. Substitution of a bound variable for a term can be computed394

by applying a term (of function type) to an argument.395

Data types with binders are also used as input to high level APIs that build terms behind396

the scenes. For example a Coq record is just an inductive type and the API to declare one397

must allow the type of a field to depend on the fields that comes before it. Note that the398

field constructor takes a coercion flag, the name of the field, its type and binds a term in399

the remaining record declaration.400

1 kind indt-decl type. % The type of an inductive type declaration
2 type record string -> term -> string -> record-decl -> indt-decl.
3 type field bool -> string -> term -> (term -> record-decl) -> record-decl.
4 type end-record record-decl.
5 ... % constructors for non-record inductive types are omitted for brevity
6

7 external pred coq.env.add-indt i:indt-decl, o:inductive.
8 external pred coq.CS.canonical-projections i:inductive, o:list (option constant).

The pred keyword documents types and modes (input or output) of the arguments of401

a predicate, while external signals that the predicate is a builtin (in other words it is402

implemented in OCaml rather than λProlog).403

We comment these two builtin predicates and the indt-decl type while looking at the404

code of declare-structure that is in charge of scripting the following Coq code:405

1 Structure type : Type := Pack { sort : Type; class : axioms sort }.

The following Elpi code builds the declaration, type checks it, adds it to the Coq406

environment and finally returns the projections for the sort and the class fields.407

1 pred declare-structure i:gref, o:term, o:term, o:term.
2

3 declare-structure ClassName Structure SortProjection ClassProjection :- std.do! [
4 StructureDeclaration =
5 record "type" {{ Type }} "Pack" (
6 field tt "sort" {{ Type }} s\
7 field ff "class" (app [global ClassName, s]) _\
8 end-record),
9 coq.typecheck-indt-decl StructureDeclaration,

10 coq.env.add-indt StructureDeclaration StructureName,
11 coq.CS.canonical-projections StructureName [some SortP, some ClassP],
12 Structure = global (indt StructureName),
13 SortProjection = global (const SortP),
14 ClassProjection = global (const ClassP),
15 ].

Note that the binder s\ at line 6 lets one mention the first field in the type of the second.408

The syntax {{ Type }} is a quotation: it lets one use the syntax of Coq to write an Elpi409

expression of type term. The API coq.CS.canonical-projections lets us find the projections410

automatically generated by Coq for a given record. The last detail worth mentioning is that411

this program makes no use of backtracking: the std.do! combinator signals that.412

In the simple case of the structure record, the number of fields, and hence the number of413

binders, is fixed. On the contrary the class record has one field per mixin and each of them414

can depend on the previous ones. In order to synthesize terms with binders in an inductive415

fashion (using a recursive predicate) λProlog lets one postulate fresh nominal constants using416

the pi operator and attach to the nominal some knowledge in the form of a clause via the =>417

operator. This process is called binder mobility: the binder is moved from the data (that we418

are building) to the program (the context of the current computation). This feature is key419

to the following code that synthesizes the declaration of the fields of the class record.420

1 pred synthesize-fields.field-for-mixin i:mixinname, o:term.
2 pred synthesize-fields i:list mixinname, i:term, o:record-decl.
3

4 synthesize-fields [] _ Decl :- Decl = end-record.
5 synthesize-fields [M|ML] T Decl :- std.do! [
6 get-mixin-modname M ModName, Name is ModName ^ "_mixin",
7 dep1 M Deps,
8 std.map Deps synthesize-fields.field-for-mixin Args,
9 Type = app[ global M, T | Args ],

10 Decl = (field ff Name Type f\ Fields f),
11 pi m\
12 synthesize-fields.field-for-mixin M m =>
13 synthesize-fields ML T (Fields m)
14 ].

The first predicate synthesize-fields.field-for-mixin is used to link a mixin to a421

nominal that corresponds to the record field for that mixin. It has no clauses in the base422

program but some clauses are added dynamically by synthesize-fields.423

The second predicate proceeds by recursion on the (topologically sorted) list of mixins,424

and terminates when the list is empty. If the list contains a mixin M then it crafts a Name for425

it (line 6), fetches its dependencies (line 7) and finds the (previously declared) record fields426

holding these mixins (line 8). The (std.map L1 P L2) predicate relates the two lists L1 and427

L2 point wise using the predicate P.428

Line 9 builds the type of the field: the mixin name applied to the type (sort) and all its429

dependencies. Note that the Fields variable, representing the declaration of the next fields,430

is under the binder for f (the current field). In order to make the recursive call under that431

binder (line 13) and recursively process ML we postulate a nominal m (line 11) that is a term432

satisfying any future dependency on the current mixin (line 12) and we replace f by m in433

Fields by writing (Fields m).434

6 Related work435

The most closely related work is the one about Packed Classes [13] on which we build. The436

main differences are that HB is a higher level language that is compiled down to (flat)437

Packed Classes. The systematic use of factories makes the user interface of a hierarchy stable438

under the insertion of structures, a property that Packed Classes lacks. Finally many of the439

intricacies of Packed Classes are hidden to the user by the compilation step, in particular440

creating all the necessary coercions and canonical structure declarations, especially in the case441

of diamonds or when merging several libraries, which used to cause the need for a posteriori442

validation of a hierarchy design [26]. It also opens the way to automatically detect and solve443

problems tied to non judgmental commutative diagrams when forgetting structure [1].444

In [7] Carette and O’Connor describe the language of Theory Presentation Combinators445

that can be used to describe a hierarchy of algebraic structures. They focus on the categorical446

semantics of the language that is built upon the category of context. They do not describe447

any actual compilation to the language of a mainstream interactive prover, indeed they claim448

their language to be mostly type theory independent. We know they considered targeting449

type theory and the language of unification hints [4] (a superset of the one of Canonical450

Structures), but we could not find any written trace of that. Language-wise they provide451

keywords such as combine and over to share (reuse) a property when defining a new structure.452

For example in order to avoid restating the commutativity property when defining Abelian453

groups they combine a commutative monoid and a group forcing the subjacent monoid454

to coincide: CommutativeGroup := combine CommutativeMonoid , Group over Monoid.455

In our language HB the same role is played by mixins. A mixin lets one write once and for456

all a property and abstract it over types and operations so that it can be reused in all the457

structures that need it. One operation HB allows for but that does not seem to be possible458

in the setting of Presentation Combinators is the one of replacing an axiom with a lemma459

and vice versa. As shown in subsection 2.1 HB supports that.460

The MMT system [22] provides a framework to describe formal languages in a logical461

framework, providing good support for binders and notations. It also provides an expressive462

module system to organize theories and express relations among them in the form of functors.463

At the time of writing it provides limited support for elaborating user input taking systematic464

advantage of the contents of the theories. The elaborator can be extended by the user writing465

Scala code, and in principle use the contents of the libraries to make sense of an incomplete466

expressions, but no higher level language or mechanism is provided.467

The library of the Mizar system features a hierarchy of algebraic structures [15]. In spite468

of lacking dependent types, Mizar provides the concept of attributed types and adjectives469

that can be used to describe the signature of structures as one would do with a dependently470

typed record and their properties as one would define a conjunctive predicate. The Mizar471

language also provides the notion of cluster that is used to link structures: by showing that472

property P implies property Q one can inform automation that structures characterized by473

P are instances of structures characterized by Q. The foundational theory of Mizar features474

an extensional notion of equality that makes it easy to share the signature or the properties475

of structures by just requiring a proof of their equivalence that is in turn used by automation476

to treat equivalent structures as equal.477

The concepts of factory and builder presented here is akin to the AbstractFactory and478

Builder pattern from "the Gang Of Four" design patterns [12] in the sense that factories are479

used to build an arbitrary number of objects (here mixin instances).480

7 Conclusion481

In this paper we design and implement HB, a high level language to describe hierarchies of482

algebraic structures. The implementation of HB is based on the Elpi extension language for483

the Coq system and is available at https://github.com/math-comp/hierarchy-builder.484

The implementation amounts to approximately a thousand lines of (commented) Elpi code485

and tree hundred lines of Coq vernacular. It took less than one month to implement HB, but486

its design took several years of attempts and fruitful discussions with the people we thank487

below.488

The HB language is only loosely tied to Coq or even Type Theory. We believe it could489

be adopted with no major change to other tools. Indeed the properties and invariants that490

link factories, mixins and classes are key to rule out meaningless or ambiguous sentences491

and are not specific to our logic setting. Moreover most logics feature packing construction492

similar to records.493

The compilation scheme we present in section 4 is tied to dependent records, that are494

available in Coq but also in other provers based on Type Theory such a Matita [3] and495

Lean [10]. We chose the flat variant of Packed Classes as the target for HB because the496

Mathematical Components library uses Packed Classes as well: As of today Packed Classes497

offer the best compromise between flexibility and performance [17, Section 8] in Coq. Of498

course one could imagine a future were other approaches such as telescopes [21] or unbundled499

classes [27] would offer equal or better performances in Coq, or in another system. It is a500

virtue of our work to provide a user language that is separate from the implementation one.501

We believe it would take a minor coding effort to retarget HB to another bundling approach.502

We leave to future work extending HB to support structures parametrized by structures503

such as the one of module over a ring. We also leave to future work the automatic synthesis504

of the notion of morphism between structures of the hierarchy.505

A Coq reference607

1 Section OperationProperties.
2

3 Variables (T : Type) (e : T) (inv : T -> T) (op : T -> T -> T) (add : T -> T -> T).
4

5 Definition left_id := forall x, op e x = x.
6 Definition right_id := forall x, op x e = x.
7

8 Definition left_inverse := forall x, op (inv x) x = e.
9 Definition right_inverse := forall x, op x (inv x) = e.

10

11 Definition commutative := forall x y, op x y = op y x.
12 Definition associative := forall x y z, op x (op y z) = op (op x y) z.
13

14 Definition left_distributive := forall x y z, op (add x y) z = add (op x z) (op y z).
15 Definition right_distributive := forall x y z, op x (add y z) = add (op x y) (op x z).
16

17 End OperationProperties.

B Proof of addrC608

1 Lemma addrC {R : Ring.type} : commutative (@add R).
2 Proof.
3 have innerC (a b : R) : a + b + (a + b) = a + a + (b + b).
4 by rewrite -[a+b]mul1r -mulrDl mulrDr !mulrDl !mul1r.
5 have addKl (a b c : R) : a + b = a + c -> b = c.
6 apply: can_inj (add a) (add (opp a)) _ _ _.
7 by move=> x; rewrite addrA addNr add0r.
8 have addKr (a b c : R) : b + a = c + a -> b = c.
9 apply: can_inj (add ^~ a) (add ^~ (opp a)) _ _ _.

10 by move=> x; rewrite /= -addrA addrN addr0.
11 move=> x y; apply: addKl (x) _ _ _; apply: addKr (y) _ _ _.
12 by rewrite -!addrA [in RHS]addrA innerC !addrA.
13 Qed.

C Full V4 code609

1 From Coq Require Import ssreflect ssrfun ZArith.
2 From HB Require Import structures.
3

4 Declare Scope hb_scope.
5 Delimit Scope hb_scope with G.
6 Open Scope hb_scope.
7

8 (* Bottom mixin in Fig. 2. *)
9 HB.mixin Record Monoid_of_Type M := {

10 zero : M;
11 add : M -> M -> M;
12 addrA : associative add;
13 add0r : left_id zero add;
14 addr0 : right_id zero add;
15 }.
16 HB.structure Definition Monoid := { M of Monoid_of_Type M }.
17 Notation "0" := zero : hb_scope.
18 Infix "+" := (@add _) : hb_scope.
19

20 (* Bottom right mixin in Fig. 2. *)
21 HB.mixin Record AbelianGroup_of_Monoid A of Monoid A := {
22 opp : A -> A;
23 addrC : commutative (add : A -> A -> A);
24 addNr : left_inverse zero opp add;
25 }.
26 HB.structure Definition AbelianGroup := { A of Monoid A & AbelianGroup_of_Monoid A }.

27 Notation "- x" := (@opp _ x) : hb_scope.
28 Notation "x - y" := (x + - y) : hb_scope.
29

30 (* Bottom left mixin in Fig. 2. *)
31 HB.mixin Record SemiRing_of_Monoid S of Monoid S := {
32 one : S;
33 mul : S -> S -> S;
34 mulrA : associative mul;
35 mul1r : left_id one mul;
36 mulr1 : right_id one mul;
37 mulrDl : left_distributive mul add;
38 mulrDr : right_distributive mul add;
39 mul0r : left_zero zero mul;
40 mulr0 : right_zero zero mul;
41 }.
42 HB.structure Definition SemiRing := { S of Monoid S & SemiRing_of_Monoid S }.
43 Notation "1" := one : hb_scope.
44 Infix "*" := (@mul _) : hb_scope.
45

46 Lemma addrN {A : AbelianGroup.type} : right_inverse (zero : A) opp add.
47 Proof. by move=> x; rewrite addrC addNr. Qed.
48

49 (* Top right factory in Fig. 2. *)
50 HB.factory Record Ring_of_AbelianGroup R of AbelianGroup R := {
51 one : R;
52 mul : R -> R -> R;
53 mulrA : associative mul;
54 mul1r : left_id one mul;
55 mulr1 : right_id one mul;
56 mulrDl : left_distributive mul add;
57 mulrDr : right_distributive mul add;
58 }.
59

60 (* Builder arrow from top right to bottom left in Fig. 2. *)
61 HB.builders Context (A : Type) (f : Ring_of_AbelianGroup A).
62

63 Fact mul0r : left_zero zero mul.
64 Proof.
65 move=> x; rewrite -[LHS]add0r addrC.
66 rewrite -{2}(addNr (mul x x)) (addrC (opp _)) addrA.
67 by rewrite -mulrDl add0r addrC addNr.
68 Qed.
69

70 Fact mulr0 : right_zero zero mul.
71 Proof.
72 move=> x; rewrite -[LHS]add0r addrC.
73 rewrite -{2}(addNr (mul x x)) (addrC (opp _)) addrA.
74 by rewrite -mulrDr add0r addrC addNr.
75 Qed.
76

77 Definition to_SemiRing_of_Monoid := SemiRing_of_Monoid.Build A _ mul mulrA
78 mul1r mulr1 mulrDl mulrDr mul0r mulr0.
79 HB.instance A to_SemiRing_of_Monoid.
80

81 HB.end.
82 HB.structure Definition Ring := { A of AbelianGroup A & Ring_of_AbelianGroup A }.
83

84 (* Top left factory in Fig. 2. *)
85 (* It is an exact copy of the bottom right mixin. *)
86 HB.factory Definition Ring_of_SemiRing R of SemiRing R := AbelianGroup_of_Monoid R.
87 (* The corresponding builder is the identity. *)
88 HB.builders Context (R : Type) (f : Ring_of_SemiRing R).
89 Definition to_AbelianGroup_of_Monoid : AbelianGroup_of_Monoid R := f.
90 HB.instance R to_AbelianGroup_of_Monoid.
91 HB.end.
92

93 (* Right-most factory in Fig. 2. *)
94 HB.factory Record Ring_of_Monoid R of Monoid R := {
95 one : R;
96 opp : R -> R;
97 mul : R -> R -> R;
98 addNr : left_inverse zero opp add;
99 addrN : right_inverse zero opp add;

100 mulrA : associative mul;
101 mul1r : left_id one mul;
102 mulr1 : right_id one mul;
103 mulrDl : left_distributive mul add;
104 mulrDr : right_distributive mul add;
105 }.
106

107 HB.builders Context (R : Type) (f : Ring_of_Monoid R).
108

109 Lemma addrC : commutative (add : R -> R -> R).
110 Proof. (* Exactly the same as in Appendix B. *)
111 have innerC (a b : R) : a + b + (a + b) = a + a + (b + b).
112 by rewrite -[a+b]mul1r -mulrDl mulrDr !mulrDl !mul1r.
113 have addKl (a b c : R) : a + b = a + c -> b = c.
114 apply: can_inj (add a) (add (opp a)) _ _ _.
115 by move=> x; rewrite addrA addNr add0r.
116 have addKr (a b c : R) : b + a = c + a -> b = c.
117 apply: can_inj (add ^~ a) (add ^~ (opp a)) _ _ _.
118 by move=> x; rewrite /= -addrA addrN addr0.
119 move=> x y; apply: addKl (x) _ _ _; apply: addKr (y) _ _ _.
120 by rewrite -!addrA [in RHS]addrA innerC !addrA.
121 Qed.
122

123 (* Builder to the bottom right mixin. *)
124 Definition to_AbelianGroup_of_Monoid :=
125 AbelianGroup_of_Monoid.Build R opp addrC addNr.
126 HB.instance R to_AbelianGroup_of_Monoid.
127

128 (* Builder to the top right factory, which is compiled to the bottom left mixin. *)
129 Definition to_Ring_of_AbelianGroup := Ring_of_AbelianGroup.Build R one mul
130 mulrA mul1r mulr1 mulrDl mulrDr.
131 HB.instance R to_Ring_of_AbelianGroup.
132

133 HB.end.

FSCD 2020


	Introduction
	HB by examples: building and evolving a hierarchy
	Evolution of the hierarchy
	The missing puzzle piece
	HB in a nutshell

	The HB language
	Mixins, factories and instances
	Classes and structures
	Automatic inference of mixins

	The target language: Coq with Packed, flat, Classes
	Describing structures with records and projections
	Multiple inheritance
	Linking structures and instances via Canonical Structures

	The implementation of HB in Coq-Elpi
	Related work
	Conclusion
	Coq reference
	Proof of addrC
	Full V4 code

